query: SELECT ?item ?itemLabel ?count ?countBeschermd ?countPubliekdomein ?countOnbekend ?countWelgeboortedatum ?beschermdPercentage ?publiekdomeinPercentage  ?onbekendPercentage ?welgeboortedatumPercentage
  WITH {
    SELECT DISTINCT ?item ?human WHERE {
      ?human wdt:P31 wd:Q5 .
      ?human wdt:P6379 ?item .
      FILTER (?item IN (wd:Q679527))
    }
  } AS %humans
  WITH {
    SELECT DISTINCT ?human ?beschermd ?publiekdomein ?onbekend ?welgeboortedatum WHERE {
      INCLUDE %humans.
      OPTIONAL {
        ?human wdt:P7763 wd:Q73555012
        BIND(1 AS ?beschermd)
      }
      OPTIONAL {
        ?human wdt:P7763 wd:Q71887839
        BIND(1 AS ?publiekdomein)
      }
      OPTIONAL {
        FILTER NOT EXISTS
         {?human wdt:P7763 []
         }
        FILTER NOT EXISTS
         {?human wdt:P569 []
         }
        BIND(1 AS ?onbekend)
      }
      OPTIONAL {
        ?human wdt:P569 ?geboortedatum
        FILTER NOT EXISTS
         {?human wdt:P7763 []}
        BIND(1 AS ?welgeboortedatum)
      }
    }
  } AS %humansWithRep
  WITH {
    SELECT ?item (COUNT(?human) AS ?count) (SUM(?beschermd) AS ?countBeschermd) (CONCAT(str(ROUND(SUM(?beschermd)* 100 / (COUNT(?human)))), "% (", str(SUM(?beschermd)), " personen)") AS ?beschermdPercentage) (SUM(?publiekdomein) AS ?countPubliekdomein) (CONCAT(str(ROUND(SUM(?publiekdomein)* 100 / (COUNT(?human)))), "% (", str(SUM(?publiekdomein)), " personen)") AS ?publiekdomeinPercentage) (SUM(?onbekend) AS ?countOnbekend) (CONCAT(str(ROUND(SUM(?onbekend)* 100 / (COUNT(?human)))), "% (", str(SUM(?onbekend)), " personen)") AS ?onbekendPercentage) (SUM(?welgeboortedatum) AS ?countWelgeboortedatum) (CONCAT(str(ROUND(SUM(?welgeboortedatum)* 100 / (COUNT(?human)))), "% (", str(SUM(?welgeboortedatum)), " personen)") AS ?welgeboortedatumPercentage) WHERE {
      INCLUDE %humans.
      INCLUDE %humansWithRep.
    }
    GROUP BY ?item
  } AS %items
  WHERE {
    INCLUDE %items.
    SERVICE wikibase:label { bd:serviceParam wikibase:language "[AUTO_LANGUAGE],en". }
  }